



Isabelle's Logics: HOL¹

Tobias Nipkow² and Lawrence C. Paulson³ and Markus Wenzel⁴

22 May 2012

¹The research has been funded by the EPSRC (grants GR/G53279, GR/H40570, GR/K57381, GR/K77051, GR/M75440), by ESPRIT (projects 3245: Logical Frameworks, and 6453: Types) and by the DFG Schwerpunktprogramm *Deduktion*.

²Institut für Informatik, Technische Universität München, nipkow@in.tum.de

³Computer Laboratory, University of Cambridge, lcp@cl.cam.ac.uk

⁴Institut für Informatik, Technische Universität München, wenzelm@in.tum.de

Abstract

This manual describes Isabelle’s formalization of Higher-Order Logic, a polymorphic version of Church’s Simple Theory of Types. HOL can be best understood as a simply-typed version of classical set theory. The monograph *Isabelle/HOL — A Proof Assistant for Higher-Order Logic* provides a gentle introduction on using Isabelle/HOL in practice.

Contents

1	Syntax definitions	1
2	Higher-Order Logic	3
2.1	Syntax	3
2.1.1	Types and overloading	3
2.1.2	Binders	6
2.1.3	The let and case constructions	7
2.2	Rules of inference	8
2.3	A formulation of set theory	9
2.3.1	Syntax of set theory	14
2.3.2	Axioms and rules of set theory	15
2.3.3	Properties of functions	18
2.4	Simplification and substitution	20
2.4.1	Case splitting	20
2.5	Types	21
2.5.1	Product and sum types	23
2.5.2	The type of natural numbers, <i>nat</i>	24
2.5.3	Numerical types and numerical reasoning	26
2.5.4	The type constructor for lists, <i>list</i>	27
2.6	Datatype definitions	30
2.6.1	Basics	31
2.6.2	Defining datatypes	35
2.7	Old-style recursive function definitions	36
2.8	Example: Cantor's Theorem	39

Syntax definitions

The syntax of each logic is presented using a context-free grammar. These grammars obey the following conventions:

- identifiers denote nonterminal symbols
- `typewriter` font denotes terminal symbols
- parentheses (...) express grouping
- constructs followed by a Kleene star, such as id^* and $(\dots)^*$ can be repeated 0 or more times
- alternatives are separated by a vertical bar, |
- the symbol for alphanumeric identifiers is *id*
- the symbol for scheme variables is *var*

To reduce the number of nonterminals and grammar rules required, Isabelle's syntax module employs **priorities**, or precedences. Each grammar rule is given by a mixfix declaration, which has a priority, and each argument place has a priority. This general approach handles infix operators that associate either to the left or to the right, as well as prefix and binding operators.

In a syntactically valid expression, an operator's arguments never involve an operator of lower priority unless brackets are used. Consider first-order logic, where \exists has lower priority than \vee , which has lower priority than \wedge . There, $P \wedge Q \vee R$ abbreviates $(P \wedge Q) \vee R$ rather than $P \wedge (Q \vee R)$. Also, $\exists x. P \vee Q$ abbreviates $\exists x. (P \vee Q)$ rather than $(\exists x. P) \vee Q$. Note especially that $P \vee (\exists x. Q)$ becomes syntactically invalid if the brackets are removed.

A **binder** is a symbol associated with a constant of type $(\sigma \Rightarrow \tau) \Rightarrow \tau'$. For instance, we may declare \forall as a binder for the constant *All*, which has type $(\alpha \Rightarrow o) \Rightarrow o$. This defines the syntax $\forall x. t$ to mean *All*($\lambda x. t$). We can also write $\forall x_1 \dots x_m. t$ to abbreviate $\forall x_1 \dots \forall x_m. t$; this is possible for any constant provided that τ and τ' are the same type. The Hilbert description operator $\varepsilon x. P x$ has type $(\alpha \Rightarrow \text{bool}) \Rightarrow \alpha$ and normally binds only one

variable. ZF's bounded quantifier $\forall x \in A . P(x)$ cannot be declared as a binder because it has type $[i, i \Rightarrow o] \Rightarrow o$. The syntax for binders allows type constraints on bound variables, as in

$$\forall(x::\alpha) (y::\beta) z::\gamma . Q(x, y, z)$$

To avoid excess detail, the logic descriptions adopt a semi-formal style. Infix operators and binding operators are listed in separate tables, which include their priorities. Grammar descriptions do not include numeric priorities; instead, the rules appear in order of decreasing priority. This should suffice for most purposes; for full details, please consult the actual syntax definitions in the `.thy` files.

Each nonterminal symbol is associated with some Isabelle type. For example, the formulae of first-order logic have type o . Every Isabelle expression of type o is therefore a formula. These include atomic formulae such as P , where P is a variable of type o , and more generally expressions such as $P(t, u)$, where P , t and u have suitable types. Therefore, ‘expression of type o ’ is listed as a separate possibility in the grammar for formulae.

Higher-Order Logic

2.1 Syntax

Figure 2.1 lists the constants (including infixes and binders), while Fig. 2.2 presents the grammar of higher-order logic. Note that $a \sim b$ is translated to $\neg(a = b)$.

! HOL has no if-and-only-if connective; logical equivalence is expressed using
 • equality. But equality has a high priority, as befitting a relation, while if-and-only-if typically has the lowest priority. Thus, $\neg\neg P = P$ abbreviates $\neg\neg(P = P)$ and not $(\neg\neg P) = P$. When using $=$ to mean logical equivalence, enclose both operands in parentheses.

2.1.1 Types and overloading

The universal type class of higher-order terms is called **term**. By default, explicit type variables have class **term**. In particular the equality symbol and quantifiers are polymorphic over class **term**.

The type of formulae, *bool*, belongs to class **term**; thus, formulae are terms. The built-in type *fun*, which constructs function types, is overloaded with arity **(term, term) term**. Thus, $\sigma \Rightarrow \tau$ belongs to class **term** if σ and τ do, allowing quantification over functions.

HOL allows new types to be declared as subsets of existing types, either using the primitive **typedef** or the more convenient **datatype** (see §2.6).

Several syntactic type classes — **plus**, **minus**, **times** and **power** — permit overloading of the operators $+$, $-$, $*$, and $^$. They are overloaded to denote the obvious arithmetic operations on types **nat**, **int** and **real**. (With the $^$ operator, the exponent always has type **nat**.) Non-arithmetic overloadings are also done: the operator $-$ can denote set difference, while $^$ can denote exponentiation of relations (iterated composition). Unary minus is also written as $-$ and is overloaded like its 2-place counterpart; it even can stand for set complement.

The constant 0 is also overloaded. It serves as the zero element of several types, of which the most important is **nat** (the natural numbers). The type

<i>name</i>	<i>meta-type</i>	<i>description</i>
Trueprop	$bool \Rightarrow prop$	coercion to <i>prop</i>
Not	$bool \Rightarrow bool$	negation (\neg)
True	$bool$	tautology (\top)
False	$bool$	absurdity (\perp)
If	$[bool, \alpha, \alpha] \Rightarrow \alpha$	conditional
Let	$[\alpha, \alpha \Rightarrow \beta] \Rightarrow \beta$	let binder

CONSTANTS

<i>symbol</i>	<i>name</i>	<i>meta-type</i>	<i>description</i>
SOME or @	Eps	$(\alpha \Rightarrow bool) \Rightarrow \alpha$	Hilbert description (ε)
ALL or !	All	$(\alpha \Rightarrow bool) \Rightarrow bool$	universal quantifier (\forall)
EX or ?	Ex	$(\alpha \Rightarrow bool) \Rightarrow bool$	existential quantifier (\exists)
EX! or ?!	Ex1	$(\alpha \Rightarrow bool) \Rightarrow bool$	unique existence ($\exists!$)
LEAST	Least	$(\alpha :: ord \Rightarrow bool) \Rightarrow \alpha$	least element

BINDERS

<i>symbol</i>	<i>meta-type</i>	<i>priority</i>	<i>description</i>
o	$[\beta \Rightarrow \gamma, \alpha \Rightarrow \beta] \Rightarrow (\alpha \Rightarrow \gamma)$	Left 55	composition (\circ)
=	$[\alpha, \alpha] \Rightarrow bool$	Left 50	equality ($=$)
<	$[\alpha :: ord, \alpha] \Rightarrow bool$	Left 50	less than ($<$)
<=	$[\alpha :: ord, \alpha] \Rightarrow bool$	Left 50	less than or equals (\leq)
&	$[bool, bool] \Rightarrow bool$	Right 35	conjunction (\wedge)
 	$[bool, bool] \Rightarrow bool$	Right 30	disjunction (\vee)
-->	$[bool, bool] \Rightarrow bool$	Right 25	implication (\rightarrow)

INFIXES

Figure 2.1: Syntax of HOL

<i>term</i>	=	expression of class <i>term</i>
		SOME <i>id</i> . <i>formula</i> @ <i>id</i> . <i>formula</i>
		let <i>id</i> = <i>term</i> ; ...; <i>id</i> = <i>term</i> in <i>term</i>
		if <i>formula</i> then <i>term</i> else <i>term</i>
		LEAST <i>id</i> . <i>formula</i>
<i>formula</i>	=	expression of type <i>bool</i>
		<i>term</i> = <i>term</i>
		<i>term</i> ~= <i>term</i>
		<i>term</i> < <i>term</i>
		<i>term</i> <= <i>term</i>
		~ <i>formula</i>
		<i>formula</i> & <i>formula</i>
		<i>formula</i> <i>formula</i>
		<i>formula</i> --> <i>formula</i>
		ALL <i>id id*</i> . <i>formula</i> ! <i>id id*</i> . <i>formula</i>
		EX <i>id id*</i> . <i>formula</i> ? <i>id id*</i> . <i>formula</i>
		EX! <i>id id*</i> . <i>formula</i> ?! <i>id id*</i> . <i>formula</i>

Figure 2.2: Full grammar for HOL

class `plus_ac0` comprises all types for which `0` and `+` satisfy the laws $x + y = y + x$, $(x + y) + z = x + (y + z)$ and $0 + x = x$. These types include the numeric ones `nat`, `int` and `real` and also multisets. The summation operator `setsum` is available for all types in this class.

Theory `Ord` defines the syntactic class `ord` of order signatures. The relations $<$ and \leq are polymorphic over this class, as are the functions `mono`, `min` and `max`, and the `LEAST` operator. `Ord` also defines a subclass `order` of `ord` which axiomatizes the types that are partially ordered with respect to \leq . A further subclass `linorder` of `order` axiomatizes linear orderings. For details, see the file `Ord.thy`.

If you state a goal containing overloaded functions, you may need to include type constraints. Type inference may otherwise make the goal more polymorphic than you intended, with confusing results. For example, the variables i, j and k in the goal $i \leq j \implies i \leq j + k$ have type $\alpha :: \{ord, plus\}$, although you may have expected them to have some numeric type, e.g. `nat`. Instead you should have stated the goal as $(i :: nat) \leq j \implies i \leq j + k$, which causes all three variables to have type `nat`.

! If resolution fails for no obvious reason, try setting `show_types` to `true`, causing Isabelle to display types of terms. Possibly set `show_sorts` to `true` as well, causing Isabelle to display type classes and sorts.

Where function types are involved, Isabelle's unification code does not guarantee to find instantiations for type variables automatically. Be prepared to use `res_inst_tac` instead of `resolve_tac`, possibly instantiating type variables. Setting `Unify.trace_types` to `true` causes Isabelle to report omitted search paths during unification.

2.1.2 Binders

Hilbert's **description** operator $\varepsilon x . P[x]$ stands for some x satisfying P , if such exists. Since all terms in HOL denote something, a description is always meaningful, but we do not know its value unless P defines it uniquely. We may write descriptions as `Eps`($\lambda x . P[x]$) or use the syntax `SOME` $x . P[x]$.

Existential quantification is defined by

$$\exists x . P \ x \equiv P(\varepsilon x . P \ x).$$

The unique existence quantifier, $\exists! x . P$, is defined in terms of \exists and \forall . An Isabelle binder, it admits nested quantifications. For instance, $\exists! x \ y . P \ x \ y$ abbreviates $\exists! x . \exists! y . P \ x \ y$; note that this does not mean that there exists a unique pair (x, y) satisfying $P \ x \ y$.

The basic Isabelle/HOL binders have two notations. Apart from the usual **ALL** and **EX** for \forall and \exists , Isabelle/HOL also supports the original notation of Gordon’s HOL system: **!** and **?**. In the latter case, the existential quantifier *must* be followed by a space; thus **?x** is an unknown, while **? x. f x=y** is a quantification. Both notations are accepted for input. The print mode “HOL” governs the output notation. If enabled (e.g. by passing option **-m HOL** to the **isabelle** executable), then **!** and **?** are displayed.

If τ is a type of class **ord**, P a formula and x a variable of type τ , then the term **LEAST** $x . P[x]$ is defined to be the least (w.r.t. \leq) x such that $P x$ holds (see Fig. 2.4). The definition uses Hilbert’s ε choice operator, so **Least** is always meaningful, but may yield nothing useful in case there is not a unique least element satisfying P .¹

All these binders have priority 10.

! The low priority of binders means that they need to be enclosed in parenthesis
 • when they occur in the context of other operations. For example, instead of $P \wedge \forall x . Q$ you need to write $P \wedge (\forall x . Q)$.

2.1.3 The let and case constructions

Local abbreviations can be introduced by a **let** construct whose syntax appears in Fig. 2.2. Internally it is translated into the constant **Let**. It can be expanded by rewriting with its definition, **Let_def**.

HOL also defines the basic syntax

$$\text{case } e \text{ of } c_1 \Rightarrow e_1 \mid \dots \mid c_n \Rightarrow e_n$$

as a uniform means of expressing **case** constructs. Therefore **case** and **of** are reserved words. Initially, this is mere syntax and has no logical meaning. By declaring translations, you can cause instances of the **case** construct to denote applications of particular case operators. This is what happens automatically for each **datatype** definition (see §2.6).

! Both **if** and **case** constructs have as low a priority as quantifiers, which re-
 • quires additional enclosing parentheses in the context of most other operations. For example, instead of $f x = \text{if} \dots \text{then} \dots \text{else} \dots$ you need to write $f x = (\text{if} \dots \text{then} \dots \text{else} \dots)$.

¹Class *ord* does not require much of its instances, so \leq need not be a well-ordering, not even an order at all!

```

refl          t = (t::'a)
subst         [| s = t; P s |] ==> P (t::'a)
ext           (!!x::'a. (f x :: 'b) = g x) ==> (%x. f x) = (%x. g x)
impI          (P ==> Q) ==> P-->Q
mp            [| P-->Q; P |] ==> Q
iff           (P-->Q) --> (Q-->P) --> (P=Q)
someI         P(x::'a) ==> P(@x. P x)
True_or_False (P=True) | (P=False)

```

Figure 2.3: The HOL rules

2.2 Rules of inference

Figure 2.3 shows the primitive inference rules of HOL, with their ML names. Some of the rules deserve additional comments:

ext expresses extensionality of functions.

iff asserts that logically equivalent formulae are equal.

someI gives the defining property of the Hilbert ε -operator. It is a form of the Axiom of Choice. The derived rule **some_equality** (see below) is often easier to use.

True_or_False makes the logic classical.²

HOL follows standard practice in higher-order logic: only a few connectives are taken as primitive, with the remainder defined obscurely (Fig. 2.4). Gordon's HOL system expresses the corresponding definitions [2, page 270] using object-equality ($=$), which is possible because equality in higher-order logic may equate formulae and even functions over formulae. But theory HOL, like all other Isabelle theories, uses meta-equality ($==$) for definitions.

! The definitions above should never be expanded and are shown for completeness only. Instead users should reason in terms of the derived rules shown below or, better still, using high-level tactics.

Some of the rules mention type variables; for example, **refl** mentions the type variable `'a`. This allows you to instantiate type variables explicitly by calling **res_inst_tac**.

²In fact, the ε -operator already makes the logic classical, as shown by Diaconescu; see Paulson [3] for details.

```

True_def   True      == ((%x::bool. x)=(%x. x))
All_def    All       == (%P. P = (%x. True))
Ex_def     Ex        == (%P. P(@x. P x))
False_def  False     == (!P. P)
not_def    not       == (%P. P-->False)
and_def    op &      == (%P Q. !R. (P-->Q-->R) --> R)
or_def     op |      == (%P Q. !R. (P-->R) --> (Q-->R) --> R)
Ex1_def    Ex1       == (%P. ? x. P x & (! y. P y --> y=x))

o_def      op o      == (%(f::'b=>'c) g x::'a. f(g x))
if_def     If P x y ==
    (%P x y. @z::'a.(P=True --> z=x) & (P=False --> z=y))
Let_def    Let s f    == f s
Least_def  Least P    == @x. P(x) & (ALL y. P(y) --> x <= y)"

```

Figure 2.4: The HOL definitions

Some derived rules are shown in Figures 2.5 and 2.6, with their ML names. These include natural rules for the logical connectives, as well as sequent-style elimination rules for conjunctions, implications, and universal quantifiers.

Note the equality rules: **ssubst** performs substitution in backward proofs, while **box_equals** supports reasoning by simplifying both sides of an equation.

The following simple tactics are occasionally useful:

strip_tac *i* applies **allI** and **impI** repeatedly to remove all outermost universal quantifiers and implications from subgoal *i*.

case_tac "*P*" *i* performs case distinction on *P* for subgoal *i*: the latter is replaced by two identical subgoals with the added assumptions *P* and $\neg P$, respectively.

smp_tac *j i* applies *j* times **spec** and then **mp** in subgoal *i*, which is typically useful when forward-chaining from an induction hypothesis. As a generalization of **mp_tac**, if there are assumptions $\forall \vec{x}. P\vec{x} \rightarrow Q\vec{x}$ and $P\vec{a}$, (\vec{x} being a vector of *j* variables) then it replaces the universally quantified implication by $Q\vec{a}$. It may instantiate unknowns. It fails if it can do nothing.

2.3 A formulation of set theory

Historically, higher-order logic gives a foundation for Russell and Whitehead's theory of classes. Let us use modern terminology and call them **sets**, but

```

sym          s=t ==> t=s
trans        [| r=s; s=t |] ==> r=t
ssubst       [| t=s; P s |] ==> P t
box_equals   [| a=b; a=c; b=d |] ==> c=d
arg_cong     x = y ==> f x = f y
fun_cong     f = g ==> f x = g x
cong         [| f = g; x = y |] ==> f x = g y
not_sym      t ~= s ==> s ~= t

```

EQUALITY

```

TrueI        True
FalseE       False ==> P

conjI        [| P; Q |] ==> P&Q
conjunct1    [| P&Q |] ==> P
conjunct2    [| P&Q |] ==> Q
conjE        [| P&Q; [| P; Q |] ==> R |] ==> R

disjI1       P ==> P|Q
disjI2       Q ==> P|Q
disjE        [| P | Q; P ==> R; Q ==> R |] ==> R

notI         (P ==> False) ==> ~ P
notE         [| ~ P; P |] ==> R
impE         [| P-->Q; P; Q ==> R |] ==> R

```

PROPOSITIONAL LOGIC

```

iffI         [| P ==> Q; Q ==> P |] ==> P=Q
iffD1        [| P=Q; P |] ==> Q
iffD2        [| P=Q; Q |] ==> P
iffE         [| P=Q; [| P --> Q; Q --> P |] ==> R |] ==> R

```

LOGICAL EQUIVALENCE

Figure 2.5: Derived rules for HOL

```

allI      (!!x. P x) ==> !x. P x
spec      !x. P x ==> P x
allE      [| !x. P x; P x ==> R |] ==> R
all_dupE  [| !x. P x; [| P x; !x. P x |] ==> R |] ==> R

exI       P x ==> ? x. P x
exE       [| ? x. P x; !!x. P x ==> Q |] ==> Q

ex1I      [| P a; !!x. P x ==> x=a |] ==> ?! x. P x
ex1E      [| ?! x. P x; !!x. [| P x; ! y. P y --> y=x |] ==> R
           |] ==> R

some_equality [| P a; !!x. P x ==> x=a |] ==> (@x. P x) = a

```

QUANTIFIERS AND DESCRIPTIONS

```

ccontr      (~P ==> False) ==> P
classical   (~P ==> P) ==> P
excluded_middle ~P | P

disjCI      (~Q ==> P) ==> P|Q
exCI        (! x. ~ P x ==> P a) ==> ? x. P x
impCE       [| P-->Q; ~ P ==> R; Q ==> R |] ==> R
iffCE       [| P=Q; [| P;Q |] ==> R; [| ~P; ~Q |] ==> R |] ==> R
notnotD     ~~P ==> P
swap        ~P ==> (~Q ==> P) ==> Q

```

CLASSICAL LOGIC

```

if_P        P ==> (if P then x else y) = x
if_not_P    ~ P ==> (if P then x else y) = y
split_if    P(if Q then x else y) = ((Q --> P x) & (~Q --> P y))

```

CONDITIONALS

Figure 2.6: More derived rules

<i>name</i>	<i>meta-type</i>	<i>description</i>
<code>{}</code>	$\alpha \text{ set}$	the empty set
<code>insert</code>	$[\alpha, \alpha \text{ set}] \Rightarrow \alpha \text{ set}$	insertion of element
<code>Collect</code>	$(\alpha \Rightarrow \text{bool}) \Rightarrow \alpha \text{ set}$	comprehension
<code>INTER</code>	$[\alpha \text{ set}, \alpha \Rightarrow \beta \text{ set}] \Rightarrow \beta \text{ set}$	intersection over a set
<code>UNION</code>	$[\alpha \text{ set}, \alpha \Rightarrow \beta \text{ set}] \Rightarrow \beta \text{ set}$	union over a set
<code>Inter</code>	$(\alpha \text{ set}) \text{ set} \Rightarrow \alpha \text{ set}$	set of sets intersection
<code>Union</code>	$(\alpha \text{ set}) \text{ set} \Rightarrow \alpha \text{ set}$	set of sets union
<code>Pow</code>	$\alpha \text{ set} \Rightarrow (\alpha \text{ set}) \text{ set}$	powerset
<code>range</code>	$(\alpha \Rightarrow \beta) \Rightarrow \beta \text{ set}$	range of a function
<code>Ball Bex</code>	$[\alpha \text{ set}, \alpha \Rightarrow \text{bool}] \Rightarrow \text{bool}$	bounded quantifiers

CONSTANTS

<i>symbol</i>	<i>name</i>	<i>meta-type</i>	<i>priority</i>	<i>description</i>
<code>INT</code>	<code>INTER1</code>	$(\alpha \Rightarrow \beta \text{ set}) \Rightarrow \beta \text{ set}$	10	intersection
<code>UN</code>	<code>UNION1</code>	$(\alpha \Rightarrow \beta \text{ set}) \Rightarrow \beta \text{ set}$	10	union

BINDERS

<i>symbol</i>	<i>meta-type</i>	<i>priority</i>	<i>description</i>
<code>‘ ‘</code>	$[\alpha \Rightarrow \beta, \alpha \text{ set}] \Rightarrow \beta \text{ set}$	Left 90	image
<code>Int</code>	$[\alpha \text{ set}, \alpha \text{ set}] \Rightarrow \alpha \text{ set}$	Left 70	intersection (\cap)
<code>Un</code>	$[\alpha \text{ set}, \alpha \text{ set}] \Rightarrow \alpha \text{ set}$	Left 65	union (\cup)
<code>:</code>	$[\alpha, \alpha \text{ set}] \Rightarrow \text{bool}$	Left 50	membership (\in)
<code><=</code>	$[\alpha \text{ set}, \alpha \text{ set}] \Rightarrow \text{bool}$	Left 50	subset (\subseteq)

INFIXES

Figure 2.7: Syntax of the theory **Set**

	<i>external</i>	<i>internal</i>	<i>description</i>
	$a \sim : b$	$\sim(a : b)$	not in
	$\{a_1, \dots\}$	$\text{insert } a_1 \dots \{\}$	finite set
	$\{x. P[x]\}$	$\text{Collect}(\lambda x. P[x])$	comprehension
	$\text{INT } x:A. B[x]$	$\text{INTER } A \lambda x. B[x]$	intersection
	$\text{UN } x:A. B[x]$	$\text{UNION } A \lambda x. B[x]$	union
$\text{ALL } x:A. P[x]$	$\text{or } ! x:A. P[x]$	$\text{Ball } A \lambda x. P[x]$	bounded \forall
$\text{EX } x:A. P[x]$	$\text{or } ? x:A. P[x]$	$\text{Bex } A \lambda x. P[x]$	bounded \exists

TRANSLATIONS

<i>term</i>	=	other terms...
		$\{\}$
		$\{ \text{term } (, \text{term})^* \}$
		$\{ \text{id} . \text{formula} \}$
		$\text{term} \text{ `` } \text{term}$
		term Int term
		term Un term
		$\text{INT } \text{id}:\text{term} . \text{term}$
		$\text{UN } \text{id}:\text{term} . \text{term}$
		$\text{INT } \text{id } \text{id}^* . \text{term}$
		$\text{UN } \text{id } \text{id}^* . \text{term}$
<i>formula</i>	=	other formulae...
		$\text{term} : \text{term}$
		$\text{term} \sim : \text{term}$
		$\text{term} \leq \text{term}$
		$\text{ALL } \text{id}:\text{term} . \text{formula} \quad \quad ! \text{id}:\text{term} . \text{formula}$
		$\text{EX } \text{id}:\text{term} . \text{formula} \quad \quad ? \text{id}:\text{term} . \text{formula}$

FULL GRAMMAR

Figure 2.8: Syntax of the theory **Set** (continued)

note that these sets are distinct from those of ZF set theory, and behave more like ZF classes.

- Sets are given by predicates over some type σ . Types serve to define universes for sets, but type-checking is still significant.
- There is a universal set (for each type). Thus, sets have complements, and may be defined by absolute comprehension.
- Although sets may contain other sets as elements, the containing set must have a more complex type.

Finite unions and intersections have the same behaviour in HOL as they do in ZF. In HOL the intersection of the empty set is well-defined, denoting the universal set for the given type.

2.3.1 Syntax of set theory

HOL's set theory is called **Set**. The type $\alpha \text{ set}$ is essentially the same as $\alpha \Rightarrow \text{bool}$. The new type is defined for clarity and to avoid complications involving function types in unification. The isomorphisms between the two types are declared explicitly. They are very natural: **Collect** maps $\alpha \Rightarrow \text{bool}$ to $\alpha \text{ set}$, while **op** : maps in the other direction (ignoring argument order).

Figure 2.7 lists the constants, infixes, and syntax translations. Figure 2.8 presents the grammar of the new constructs. Infix operators include union and intersection ($A \cup B$ and $A \cap B$), the subset and membership relations, and the image operator $'\cdot'$. Note that $a \sim b$ is translated to $\neg(a \in b)$.

The $\{a_1, \dots\}$ notation abbreviates finite sets constructed in the obvious manner using **insert** and **{}**:

$$\{a, b, c\} \equiv \text{insert } a (\text{insert } b (\text{insert } c \{\}))$$

The set $\{x. P[x]\}$ consists of all x (of suitable type) that satisfy $P[x]$, where $P[x]$ is a formula that may contain free occurrences of x . This syntax expands to **Collect**($\lambda x. P[x]$). It defines sets by absolute comprehension, which is impossible in ZF; the type of x implicitly restricts the comprehension.

The set theory defines two **bounded quantifiers**:

$$\begin{aligned} \forall x \in A. P[x] & \text{ abbreviates } \forall x. x \in A \rightarrow P[x] \\ \exists x \in A. P[x] & \text{ abbreviates } \exists x. x \in A \wedge P[x] \end{aligned}$$

<code>mem_Collect_eq</code>	$(a : \{x. P\ x\}) = P\ a$	
<code>Collect_mem_eq</code>	$\{x. x:A\} = A$	
<code>empty_def</code>	$\{\}$	$== \{x. \text{False}\}$
<code>insert_def</code>	$\text{insert } a\ B$	$== \{x. x=a\} \cup B$
<code>Ball_def</code>	$\text{Ball } A\ P$	$== !\ x. x:A \rightarrow P\ x$
<code>Bex_def</code>	$\text{Bex } A\ P$	$== ?\ x. x:A \ \&\ P\ x$
<code>subset_def</code>	$A \leq B$	$== !\ x:A. x:B$
<code>Un_def</code>	$A \cup B$	$== \{x. x:A \mid x:B\}$
<code>Int_def</code>	$A \cap B$	$== \{x. x:A \ \&\ x:B\}$
<code>set_diff_def</code>	$A - B$	$== \{x. x:A \ \&\ x \sim B\}$
<code>Compl_def</code>	$\neg A$	$== \{x. \sim x:A\}$
<code>INTER_def</code>	$\text{INTER } A\ B$	$== \{y. !\ x:A. y: B\ x\}$
<code>UNION_def</code>	$\text{UNION } A\ B$	$== \{y. ?\ x:A. y: B\ x\}$
<code>INTER1_def</code>	$\text{INTER1 } B$	$== \text{INTER } \{x. \text{True}\} B$
<code>UNION1_def</code>	$\text{UNION1 } B$	$== \text{UNION } \{x. \text{True}\} B$
<code>Inter_def</code>	$\text{Inter } S$	$== (\text{INT } x:S. x)$
<code>Union_def</code>	$\text{Union } S$	$== (\text{UN } x:S. x)$
<code>Pow_def</code>	$\text{Pow } A$	$== \{B. B \leq A\}$
<code>image_def</code>	$f\ 'A$	$== \{y. ?\ x:A. y=f\ x\}$
<code>range_def</code>	$\text{range } f$	$== \{y. ?\ x. y=f\ x\}$

Figure 2.9: Rules of the theory **Set**

The constants `Ball` and `Bex` are defined accordingly. Instead of `Ball` $A\ P$ and `Bex` $A\ P$ we may write $\text{ALL } x:A. P[x]$ and $\text{EX } x:A. P[x]$. The original notation of Gordon's HOL system is supported as well: $!$ and $?$.

Unions and intersections over sets, namely $\bigcup_{x \in A} B[x]$ and $\bigcap_{x \in A} B[x]$, are written `UN` $x:A. B[x]$ and `INT` $x:A. B[x]$.

Unions and intersections over types, namely $\bigcup_x B[x]$ and $\bigcap_x B[x]$, are written `UN` $x. B[x]$ and `INT` $x. B[x]$. They are equivalent to the previous union and intersection operators when A is the universal set.

The operators $\bigcup A$ and $\bigcap A$ act upon sets of sets. They are not binders, but are equal to $\bigcup_{x \in A} x$ and $\bigcap_{x \in A} x$, respectively.

2.3.2 Axioms and rules of set theory

Figure 2.9 presents the rules of theory **Set**. The axioms `mem_Collect_eq` and `Collect_mem_eq` assert that the functions `Collect` and `op` : are isomorphisms. Of course, `op` : also serves as the membership relation.

All the other axioms are definitions. They include the empty set, bounded quantifiers, unions, intersections, complements and the subset relation. They also include straightforward constructions on functions: `image` (`'`) and

CollectI	$[P a] \Rightarrow a : \{x. P x\}$
CollectD	$[a : \{x. P x\}] \Rightarrow P a$
CollectE	$[a : \{x. P x\}; P a \Rightarrow W] \Rightarrow W$
ballI	$[!!x. x:A \Rightarrow P x] \Rightarrow ! x:A. P x$
bspec	$[! x:A. P x; x:A] \Rightarrow P x$
ballE	$[! x:A. P x; P x \Rightarrow Q; \sim x:A \Rightarrow Q] \Rightarrow Q$
bexI	$[P x; x:A] \Rightarrow ? x:A. P x$
bexCI	$[! x:A. \sim P x \Rightarrow P a; a:A] \Rightarrow ? x:A. P x$
bexE	$[? x:A. P x; !!x. [x:A; P x] \Rightarrow Q] \Rightarrow Q$

COMPREHENSION AND BOUNDED QUANTIFIERS

subsetI	$(!!x. x:A \Rightarrow x:B) \Rightarrow A \leq B$
subsetD	$[A \leq B; c:A] \Rightarrow c:B$
subsetCE	$[A \leq B; \sim (c:A) \Rightarrow P; c:B \Rightarrow P] \Rightarrow P$
subset_refl	$A \leq A$
subset_trans	$[A \leq B; B \leq C] \Rightarrow A \leq C$
equalityI	$[A \leq B; B \leq A] \Rightarrow A = B$
equalityD1	$A = B \Rightarrow A \leq B$
equalityD2	$A = B \Rightarrow B \leq A$
equalityE	$[A = B; [A \leq B; B \leq A] \Rightarrow P] \Rightarrow P$
equalityCE	$[A = B; [c:A; c:B] \Rightarrow P;$ $[\sim c:A; \sim c:B] \Rightarrow P$ $] \Rightarrow P$

THE SUBSET AND EQUALITY RELATIONS

Figure 2.10: Derived rules for set theory

```

emptyE   a : {} ==> P

insertI1 a : insert a B
insertI2 a : B ==> a : insert b B
insertE  [| a : insert b A;  a=b ==> P;  a:A ==> P |] ==> P

ComplI   [| c:A ==> False |] ==> c : -A
ComplD   [| c : -A |] ==> ~ c:A

UnI1     c:A ==> c : A Un B
UnI2     c:B ==> c : A Un B
UnCI     (~c:B ==> c:A) ==> c : A Un B
UnE      [| c : A Un B;  c:A ==> P;  c:B ==> P |] ==> P

IntI     [| c:A;  c:B |] ==> c : A Int B
IntD1    c : A Int B ==> c:A
IntD2    c : A Int B ==> c:B
IntE     [| c : A Int B;  [| c:A; c:B |] ==> P |] ==> P

UN_I     [| a:A;  b: B a |] ==> b: (UN x:A. B x)
UN_E     [| b: (UN x:A. B x);  !!x.[| x:A;  b:B x |] ==> R |] ==> R

INT_I     (!!x. x:A ==> b: B x) ==> b : (INT x:A. B x)
INT_D     [| b: (INT x:A. B x);  a:A |] ==> b: B a
INT_E     [| b: (INT x:A. B x);  b: B a ==> R;  ~ a:A ==> R |] ==> R

UnionI   [| X:C;  A:X |] ==> A : Union C
UnionE   [| A : Union C;  !!X.[| A:X;  X:C |] ==> R |] ==> R

InterI   [| !!X. X:C ==> A:X |] ==> A : Inter C
InterD   [| A : Inter C;  X:C |] ==> A:X
InterE   [| A : Inter C;  A:X ==> R;  ~ X:C ==> R |] ==> R

PowI     A<=B ==> A: Pow B
PowD     A: Pow B ==> A<=B

imageI   [| x:A |] ==> f x : f``A
imageE   [| b : f``A;  !!x.[| b=f x;  x:A |] ==> P |] ==> P

rangeI   f x : range f
rangeE   [| b : range f;  !!x.[| b=f x |] ==> P |] ==> P

```

Figure 2.11: Further derived rules for set theory

Union_upper	$B:A \implies B \leq \text{Union } A$
Union_least	$[\text{!!}X. X:A \implies X \leq C \] \implies \text{Union } A \leq C$
Inter_lower	$B:A \implies \text{Inter } A \leq B$
Inter_greatest	$[\text{!!}X. X:A \implies C \leq X \] \implies C \leq \text{Inter } A$
Un_upper1	$A \leq A \text{ Un } B$
Un_upper2	$B \leq A \text{ Un } B$
Un_least	$[A \leq C; B \leq C \] \implies A \text{ Un } B \leq C$
Int_lower1	$A \text{ Int } B \leq A$
Int_lower2	$A \text{ Int } B \leq B$
Int_greatest	$[C \leq A; C \leq B \] \implies C \leq A \text{ Int } B$

Figure 2.12: Derived rules involving subsets

range.

Figures 2.10 and 2.11 present derived rules. Most are obvious and resemble rules of Isabelle’s ZF set theory. Certain rules, such as `subsetCE`, `bexCI` and `UnCI`, are designed for classical reasoning; the rules `subsetD`, `bexI`, `Un1` and `Un2` are not strictly necessary but yield more natural proofs. Similarly, `equalityCE` supports classical reasoning about extensionality, after the fashion of `iffCE`. See the file `HOL/Set.ML` for proofs pertaining to set theory.

Figure 2.12 presents lattice properties of the subset relation. Unions form least upper bounds; non-empty intersections form greatest lower bounds. Reasoning directly about subsets often yields clearer proofs than reasoning about the membership relation. See the file `HOL/subset.ML`.

Figure 2.13 presents many common set equalities. They include commutative, associative and distributive laws involving unions, intersections and complements. For a complete listing see the file `HOL/equalities.ML`.

- ! Blast_tac proves many set-theoretic theorems automatically. Hence you seldom need to refer to the theorems above.

2.3.3 Properties of functions

Figure 2.14 presents a theory of simple properties of functions. Note that `inv f` uses Hilbert’s ε to yield an inverse of f . See the file `HOL/Fun.ML` for a complete listing of the derived rules. Reasoning about function composition (the operator `o`) and the predicate `surj` is done simply by expanding the definitions.

Int_absorb	$A \text{ Int } A = A$
Int_commute	$A \text{ Int } B = B \text{ Int } A$
Int_assoc	$(A \text{ Int } B) \text{ Int } C = A \text{ Int } (B \text{ Int } C)$
Int_Un_distrib	$(A \text{ Un } B) \text{ Int } C = (A \text{ Int } C) \text{ Un } (B \text{ Int } C)$
Un_absorb	$A \text{ Un } A = A$
Un_commute	$A \text{ Un } B = B \text{ Un } A$
Un_assoc	$(A \text{ Un } B) \text{ Un } C = A \text{ Un } (B \text{ Un } C)$
Un_Int_distrib	$(A \text{ Int } B) \text{ Un } C = (A \text{ Un } C) \text{ Int } (B \text{ Un } C)$
Compl_disjoint	$A \text{ Int } (\neg A) = \{x. \text{False}\}$
Compl_partition	$A \text{ Un } (\neg A) = \{x. \text{True}\}$
double_complement	$\neg(\neg A) = A$
Compl_Un	$\neg(A \text{ Un } B) = (\neg A) \text{ Int } (\neg B)$
Compl_Int	$\neg(A \text{ Int } B) = (\neg A) \text{ Un } (\neg B)$
Union_Un_distrib	$\text{Union}(A \text{ Un } B) = (\text{Union } A) \text{ Un } (\text{Union } B)$
Int_Union	$A \text{ Int } (\text{Union } B) = (\text{UN } C:B. A \text{ Int } C)$
Inter_Un_distrib	$\text{Inter}(A \text{ Un } B) = (\text{Inter } A) \text{ Int } (\text{Inter } B)$
Un_Inter	$A \text{ Un } (\text{Inter } B) = (\text{INT } C:B. A \text{ Un } C)$

Figure 2.13: Set equalities

<i>name</i>	<i>meta-type</i>	<i>description</i>
inj surj	$(\alpha \Rightarrow \beta) \Rightarrow \text{bool}$	injective/surjective
inj_on	$[\alpha \Rightarrow \beta, \alpha \text{ set}] \Rightarrow \text{bool}$	injective over subset
inv	$(\alpha \Rightarrow \beta) \Rightarrow (\beta \Rightarrow \alpha)$	inverse function
inj_def	inj f	$== ! x y. f x = f y \rightarrow x = y$
surj_def	surj f	$== ! y. ? x. y = f x$
inj_on_def	inj_on f A	$== !x:A. !y:A. f x = f y \rightarrow x = y$
inv_def	inv f	$== (\%y. @x. f(x) = y)$

Figure 2.14: Theory Fun

There is also a large collection of monotonicity theorems for constructions on sets in the file `HOL/mono.ML`.

2.4 Simplification and substitution

Simplification tactics such as `Asm_simp_tac` and `Full_simp_tac` use the default simpset (`simpset()`), which works for most purposes. A quite minimal simplification set for higher-order logic is `HOL_ss`; even more frugal is `HOL_basic_ss`. Equality (`=`), which also expresses logical equivalence, may be used for rewriting. See the file `HOL/simpdata.ML` for a complete listing of the basic simplification rules.

See the *Reference Manual* for details of substitution and simplification.

! Reducing $a = b \wedge P(a)$ to $a = b \wedge P(b)$ is sometimes advantageous. The left part of a conjunction helps in simplifying the right part. This effect is not available by default: it can be slow. It can be obtained by including `conj_cong` in a simpset, `addcongs [conj_cong]`.

! By default only the condition of an `if` is simplified but not the `then` and `else` parts. Of course the latter are simplified once the condition simplifies to `True` or `False`. To ensure full simplification of all parts of a conditional you must remove `if_weak_cong` from the simpset, `delcongs [if_weak_cong]`.

If the simplifier cannot use a certain rewrite rule — either because of nontermination or because its left-hand side is too flexible — then you might try `stac`:

`stac thm i`, where *thm* is of the form $lhs = rhs$, replaces in subgoal *i* instances of *lhs* by corresponding instances of *rhs*. In case of multiple instances of *lhs* in subgoal *i*, backtracking may be necessary to select the desired ones.

If *thm* is a conditional equality, the instantiated condition becomes an additional (first) subgoal.

HOL provides the tactic `hyp_subst_tac`, which substitutes for an equality throughout a subgoal and its hypotheses. This tactic uses HOL's general substitution rule.

2.4.1 Case splitting

HOL also provides convenient means for case splitting during rewriting. Goals containing a subterm of the form `if b then...else...` often require

a case distinction on b . This is expressed by the theorem `split_if`:

$$?P(\text{if } ?b \text{ then } ?x \text{ else } ?y) = ((?b \rightarrow ?P(?x)) \wedge (\neg ?b \rightarrow ?P(?y))) \quad (*)$$

For example, a simple instance of $(*)$ is

$$x \in (\text{if } x \in A \text{ then } A \text{ else } \{x\}) = ((x \in A \rightarrow x \in A) \wedge (x \notin A \rightarrow x \in \{x\}))$$

Because $(*)$ is too general as a rewrite rule for the simplifier (the left-hand side is not a higher-order pattern in the sense of the *Reference Manual*), there is a special infix function `addsplits` of type `simpset * thm list -> simpset` (analogous to `addsims`) that adds rules such as $(*)$ to a simpset, as in

```
by(simp_tac (simpset() addsplits [split_if]) 1);
```

The effect is that after each round of simplification, one occurrence of `if` is split according to `split_if`, until all occurrences of `if` have been eliminated.

It turns out that using `split_if` is almost always the right thing to do. Hence `split_if` is already included in the default simpset. If you want to delete it from a simpset, use `delsplits`, which is the inverse of `addsplits`:

```
by(simp_tac (simpset() delsplits [split_if]) 1);
```

In general, `addsplits` accepts rules of the form

$$?P(c \ ?x_1 \ \dots \ ?x_n) = rhs$$

where c is a constant and rhs is arbitrary. Note that $(*)$ is of the right form because internally the left-hand side is $?P(\text{If } ?b \ ?x \ ?y)$. Important further examples are splitting rules for `case` expressions (see §2.5.4 and §2.6.1).

Analogous to `Addsims` and `Delsims`, there are also imperative versions of `addsplits` and `delsplits`

```
Addsplits: thm list -> unit
Delsplits: thm list -> unit
```

for adding splitting rules to, and deleting them from the current simpset.

2.5 Types

This section describes HOL's basic predefined types ($\alpha \times \beta$, $\alpha + \beta$, `nat` and α *list*) and ways for introducing new types in general. The most important type construction, the `datatype`, is treated separately in §2.6.

<i>symbol</i>	<i>meta-type</i>	<i>description</i>
Pair	$[\alpha, \beta] \Rightarrow \alpha \times \beta$	ordered pairs (a, b)
fst	$\alpha \times \beta \Rightarrow \alpha$	first projection
snd	$\alpha \times \beta \Rightarrow \beta$	second projection
split	$[[\alpha, \beta] \Rightarrow \gamma, \alpha \times \beta] \Rightarrow \gamma$	generalized projection
Sigma	$[\alpha \text{ set}, \alpha \Rightarrow \beta \text{ set}] \Rightarrow (\alpha \times \beta) \text{ set}$	general sum of sets
Sigma_def Sigma A B == UN x:A. UN y:B x. {(x,y)}		
Pair_eq ((a,b) = (a',b')) = (a=a' & b=b')		
Pair_inject [(a, b) = (a',b'); [a=a'; b=b'] ==> R] ==> R		
PairE [!!x y. p = (x,y) ==> Q] ==> Q		
fst_conv fst (a,b) = a		
snd_conv snd (a,b) = b		
surjective_pairing p = (fst p, snd p)		
split split c (a,b) = c a b		
split_split R(split c p) = (! x y. p = (x,y) --> R(c x y))		
SigmaI [a:A; b:B a] ==> (a,b) : Sigma A B		
SigmaE [c:Sigma A B; !!x y. [x:A; y:B x; c=(x,y)] ==> P] ==> P		

Figure 2.15: Type $\alpha \times \beta$

2.5.1 Product and sum types

Theory `Prod` (Fig. 2.15) defines the product type $\alpha \times \beta$, with the ordered pair syntax (a, b) . General tuples are simulated by pairs nested to the right:

external	internal
$\tau_1 \times \dots \times \tau_n$	$\tau_1 \times (\dots (\tau_{n-1} \times \tau_n) \dots)$
(t_1, \dots, t_n)	$(t_1, (\dots (t_{n-1}, t_n) \dots))$

In addition, it is possible to use tuples as patterns in abstractions:

$\%(x, y). t$ stands for `split(%x y. t)`

Nested patterns are also supported. They are translated stepwise:

$$\begin{aligned} \%(x, y, z). t &\rightsquigarrow \%(x, (y, z)). t \\ &\rightsquigarrow \text{split}(\%x. \%(y, z). t) \\ &\rightsquigarrow \text{split}(\%x. \text{split}(\%y z. t)) \end{aligned}$$

The reverse translation is performed upon printing.

! The translation between patterns and `split` is performed automatically by the parser and printer. Thus the internal and external form of a term may differ, which can affect proofs. For example the term $\%(x, y). (y, x)(a, b)$ requires the theorem `split` (which is in the default simpset) to rewrite to (b, a) .

In addition to explicit λ -abstractions, patterns can be used in any variable binding construct which is internally described by a λ -abstraction. Some important examples are

Let: `let pattern = t in u`

Quantifiers: `ALL pattern: A. P`

Choice: `SOME pattern. P`

Set operations: `UN pattern: A. B`

Sets: `{pattern. P}`

There is a simple tactic which supports reasoning about patterns:

`split_all_tac i` replaces in subgoal i all $!!$ -quantified variables of product type by individual variables for each component. A simple example:

```
1. !!p. (x, y, z). (x, y, z) p = p
by(split_all_tac 1);
1. !!x xa ya. (x, y, z). (x, y, z) (x, xa, ya) = (x, xa, ya)
```

<i>symbol</i>	<i>meta-type</i>	<i>description</i>
<code>Inl</code>	$\alpha \Rightarrow \alpha + \beta$	first injection
<code>Inr</code>	$\beta \Rightarrow \alpha + \beta$	second injection
<code>sum_case</code>	$[\alpha \Rightarrow \gamma, \beta \Rightarrow \gamma, \alpha + \beta] \Rightarrow \gamma$	conditional
<code>Inl_not_Inr</code>	<code>Inl a ~ = Inr b</code>	
<code>inj_Inl</code>	<code>inj Inl</code>	
<code>inj_Inr</code>	<code>inj Inr</code>	
<code>sumE</code>	<code>[!!x. P(Inl x); !!y. P(Inr y)] ==> P s</code>	
<code>sum_case_Inl</code>	<code>sum_case f g (Inl x) = f x</code>	
<code>sum_case_Inr</code>	<code>sum_case f g (Inr x) = g x</code>	
<code>surjective_sum</code>	<code>sum_case (%x. f(Inl x)) (%y. f(Inr y)) s = f s</code>	
<code>sum.split_case</code>	<code>R(sum_case f g s) = ((! x. s = Inl(x) --> R(f(x))) & (! y. s = Inr(y) --> R(g(y))))</code>	

Figure 2.16: Type $\alpha + \beta$

Theory `Prod` also introduces the degenerate product type `unit` which contains only a single element named `()` with the property

`unit_eq` `u = ()`

Theory `Sum` (Fig. 2.16) defines the sum type $\alpha + \beta$ which associates to the right and has a lower priority than `*`: $\tau_1 + \tau_2 + \tau_3 * \tau_4$ means $\tau_1 + (\tau_2 + (\tau_3 * \tau_4))$.

The definition of products and sums in terms of existing types is not shown. The constructions are fairly standard and can be found in the respective theory files. Although the sum and product types are constructed manually for foundational reasons, they are represented as actual datatypes later.

2.5.2 The type of natural numbers, *nat*

The theory `Nat` defines the natural numbers in a roundabout but traditional way. The axiom of infinity postulates a type *ind* of individuals, which is non-empty and closed under an injective operation. The natural numbers are inductively generated by choosing an arbitrary individual for 0 and using the injective operation to take successors. This is a least fixedpoint construction.

Type *nat* is an instance of class `ord`, which makes the overloaded functions of this class (especially `<` and `<=`, but also `min`, `max` and `LEAST`) available on

<i>symbol</i>	<i>meta-type</i>	<i>priority</i>	<i>description</i>
0	α		zero
Suc	$\text{nat} \Rightarrow \text{nat}$		successor function
*	$[\alpha, \alpha] \Rightarrow \alpha$	Left 70	multiplication
div	$[\alpha, \alpha] \Rightarrow \alpha$	Left 70	division
mod	$[\alpha, \alpha] \Rightarrow \alpha$	Left 70	modulus
dvd	$[\alpha, \alpha] \Rightarrow \text{bool}$	Left 70	“divides” relation
+	$[\alpha, \alpha] \Rightarrow \alpha$	Left 65	addition
-	$[\alpha, \alpha] \Rightarrow \alpha$	Left 65	subtraction

CONSTANTS AND INFIXES

nat_induct	$[P\ 0; \text{!!}n. P\ n \Rightarrow P(\text{Suc}\ n) \] \Rightarrow P\ n$
Suc_not_Zero	$\text{Suc}\ m \sim = 0$
inj_Suc	inj Suc
n_not_Suc_n	$n \sim = \text{Suc}\ n$

BASIC PROPERTIES

Figure 2.17: The type of natural numbers, *nat*

	$0+n = n$
	$(\text{Suc}\ m)+n = \text{Suc}(m+n)$
	$m-0 = m$
	$0-n = n$
	$\text{Suc}(m)-\text{Suc}(n) = m-n$
	$0*n = 0$
	$\text{Suc}(m)*n = n + m*n$
mod_less	$m < n \Rightarrow m \bmod n = m$
mod_geq	$[0 < n; \sim m < n \] \Rightarrow m \bmod n = (m-n) \bmod n$
div_less	$m < n \Rightarrow m \text{ div } n = 0$
div_geq	$[0 < n; \sim m < n \] \Rightarrow m \text{ div } n = \text{Suc}((m-n) \text{ div } n)$

Figure 2.18: Recursion equations for the arithmetic operators

nat. Theory **Nat** also shows that \leq is a linear order, so *nat* is also an instance of class **linorder**.

Theory **NatArith** develops arithmetic on the natural numbers. It defines addition, multiplication and subtraction. Theory **Divides** defines division, remainder and the “divides” relation. The numerous theorems proved include commutative, associative, distributive, identity and cancellation laws. See Figs. 2.17 and 2.18. The recursion equations for the operators $+$, $-$ and $*$ on *nat* are part of the default simpset.

Functions on *nat* can be defined by primitive or well-founded recursion; see §2.7. A simple example is addition. Here, **op** $+$ is the name of the infix operator $+$, following the standard convention.

```
primrec
  "0 + n = n"
  "Suc m + n = Suc (m + n)"
```

There is also a **case**-construct of the form

```
case e of 0 => a | Suc m => b
```

Note that Isabelle insists on precisely this format; you may not even change the order of the two cases. Both **primrec** and **case** are realized by a recursion operator **nat_rec**, which is available because *nat* is represented as a datatype.

Tactic **induct_tac** "*n*" *i* performs induction on variable *n* in subgoal *i* using theorem **nat_induct**. There is also the derived theorem **less_induct**:

```
[| !!n. [| ! m. m < n --> P m |] ==> P n |] ==> P n
```

2.5.3 Numerical types and numerical reasoning

The integers (type **int**) are also available in HOL, and the reals (type **real**) are available in the logic image **HOL-Complex**. They support the expected operations of addition ($+$), subtraction ($-$) and multiplication ($*$), and much else. Type **int** provides the **div** and **mod** operators, while type **real** provides real division and other operations. Both types belong to class **linorder**, so they inherit the relational operators and all the usual properties of linear orderings. For full details, please survey the theories in subdirectories **Integ**, **Real**, and **Complex**.

All three numeric types admit numerals of the form $sd \dots d$, where *s* is an optional minus sign and $d \dots d$ is a string of digits. Numerals are represented internally by a datatype for binary notation, which allows numerical calculations to be performed by rewriting. For example, the integer division of 54342339 by 3452 takes about five seconds. By default, the simplifier cancels like terms on the opposite sites of relational operators (reducing $z+x < x+y$ to

$z < y$, for instance. The simplifier also collects like terms, replacing $x+y+x*3$ by $4*x+y$.

Sometimes numerals are not wanted, because for example $n+3$ does not match a pattern of the form $\text{Suc } k$. You can re-arrange the form of an arithmetic expression by proving (via `subgoal_tac`) a lemma such as $n+3 = \text{Suc } (\text{Suc } (\text{Suc } n))$. As an alternative, you can disable the fancier simplifications by using a basic simpset such as `HOL_ss` rather than the default one, `simpset()`.

Reasoning about arithmetic inequalities can be tedious. Fortunately, HOL provides a decision procedure for *linear arithmetic*: formulae involving addition and subtraction. The simplifier invokes a weak version of this decision procedure automatically. If this is not sufficient, you can invoke the full procedure `Lin_Arith.tac` explicitly. It copes with arbitrary formulae involving $=$, $<$, $<=$, $+$, $-$, Suc , min , max and numerical constants. Other subterms are treated as atomic, while subformulae not involving numerical types are ignored. Quantified subformulae are ignored unless they are positive universal or negative existential. The running time is exponential in the number of occurrences of min , max , and $-$ because they require case distinctions. If k is a numeral, then $\text{div } k$, $\text{mod } k$ and $k \text{ dvd}$ are also supported. The former two are eliminated by case distinctions, again blowing up the running time. If the formula involves explicit quantifiers, `Lin_Arith.tac` may take super-exponential time and space.

If `Lin_Arith.tac` fails, try to find relevant arithmetic results in the library. The theories `Nat` and `NatArith` contain theorems about $<$, $<=$, $+$, $-$ and $*$. Theory `Divides` contains theorems about div and mod . Use Proof General's *find* button (or other search facilities) to locate them.

2.5.4 The type constructor for lists, *list*

Figure 2.19 presents the theory `List`: the basic list operations with their types and syntax. Type $\alpha \text{ list}$ is defined as a `datatype` with the constructors `[]` and `#`. As a result the generic structural induction and case analysis tactics `induct_tac` and `cases_tac` also become available for lists. A `case` construct of the form

$$\text{case } e \text{ of } [] \Rightarrow a \mid x\#xs \Rightarrow b$$

is defined by translation. For details see §2.6. There is also a case splitting rule `split_list_case`

$$\begin{aligned} P(\text{case } e \text{ of } [] \Rightarrow a \mid x\#xs \Rightarrow f \ x \ xs) = \\ ((e = [] \rightarrow P(a)) \wedge (\forall x \ xs. e = x\#xs \rightarrow P(f \ x \ xs))) \end{aligned}$$

<i>symbol</i>	<i>meta-type</i>	<i>priority</i>	<i>description</i>
<code>[]</code>	$\alpha \text{ list}$		empty list
<code>#</code>	$[\alpha, \alpha \text{ list}] \Rightarrow \alpha \text{ list}$	Right 65	list constructor
<code>null</code>	$\alpha \text{ list} \Rightarrow \text{bool}$		emptiness test
<code>hd</code>	$\alpha \text{ list} \Rightarrow \alpha$		head
<code>tl</code>	$\alpha \text{ list} \Rightarrow \alpha \text{ list}$		tail
<code>last</code>	$\alpha \text{ list} \Rightarrow \alpha$		last element
<code>butlast</code>	$\alpha \text{ list} \Rightarrow \alpha \text{ list}$		drop last element
<code>@</code>	$[\alpha \text{ list}, \alpha \text{ list}] \Rightarrow \alpha \text{ list}$	Left 65	append
<code>map</code>	$(\alpha \Rightarrow \beta) \Rightarrow (\alpha \text{ list} \Rightarrow \beta \text{ list})$		apply to all
<code>filter</code>	$(\alpha \Rightarrow \text{bool}) \Rightarrow (\alpha \text{ list} \Rightarrow \alpha \text{ list})$		filter functional
<code>set</code>	$\alpha \text{ list} \Rightarrow \alpha \text{ set}$		elements
<code>mem</code>	$\alpha \Rightarrow \alpha \text{ list} \Rightarrow \text{bool}$	Left 55	membership
<code>foldl</code>	$(\beta \Rightarrow \alpha \Rightarrow \beta) \Rightarrow \beta \Rightarrow \alpha \text{ list} \Rightarrow \beta$		iteration
<code>concat</code>	$(\alpha \text{ list}) \text{ list} \Rightarrow \alpha \text{ list}$		concatenation
<code>rev</code>	$\alpha \text{ list} \Rightarrow \alpha \text{ list}$		reverse
<code>length</code>	$\alpha \text{ list} \Rightarrow \text{nat}$		length
<code>!</code>	$\alpha \text{ list} \Rightarrow \text{nat} \Rightarrow \alpha$	Left 100	indexing
<code>take, drop</code>	$\text{nat} \Rightarrow \alpha \text{ list} \Rightarrow \alpha \text{ list}$		take/drop a prefix
<code>takeWhile,</code> <code>dropWhile</code>	$(\alpha \Rightarrow \text{bool}) \Rightarrow \alpha \text{ list} \Rightarrow \alpha \text{ list}$		take/drop a prefix

CONSTANTS AND INFIXES

<i>external</i>	<i>internal</i>	<i>description</i>
$[x_1, \dots, x_n]$	$x_1 \# \dots \# x_n \# []$	finite list
$[x:l. P]$	<code>filter</code> $(\lambda x.P) \ l$	list comprehension

TRANSLATIONS

Figure 2.19: The theory `List`


```

null [] = True
null (x#xs) = False

hd (x#xs) = x

tl (x#xs) = xs
tl [] = []

[] @ ys = ys
(x#xs) @ ys = x # xs @ ys

set [] = {}
set (x#xs) = insert x (set xs)

x mem [] = False
x mem (y#ys) = (if y=x then True else x mem ys)

concat([]) = []
concat(x#xs) = x @ concat(xs)

rev([]) = []
rev(x#xs) = rev(xs) @ [x]

length([]) = 0
length(x#xs) = Suc(length(xs))

xs!0 = hd xs
xs!(Suc n) = (tl xs)!n

```

Figure 2.20: Simple list processing functions

```

map f [] = []
map f (x#xs) = f x # map f xs

filter P [] = []
filter P (x#xs) = (if P x then x#filter P xs else filter P xs)

foldl f a [] = a
foldl f a (x#xs) = foldl f (f a x) xs

take n [] = []
take n (x#xs) = (case n of 0 => [] | Suc(m) => x # take m xs)

drop n [] = []
drop n (x#xs) = (case n of 0 => x#xs | Suc(m) => drop m xs)

takeWhile P [] = []
takeWhile P (x#xs) = (if P x then x#takeWhile P xs else [])

dropWhile P [] = []
dropWhile P (x#xs) = (if P x then dropWhile P xs else xs)

```

Figure 2.21: Further list processing functions

which can be fed to `addsplits` just like `split_if` (see §2.4.1).

`List` provides a basic library of list processing functions defined by primitive recursion. The recursion equations are shown in Figs. 2.20 and 2.21.

2.6 Datatype definitions

Inductive datatypes, similar to those of ML, frequently appear in applications of Isabelle/HOL. In principle, such types could be defined by hand via `typedef`, but this would be far too tedious. The `datatype` definition package of Isabelle/HOL (cf. [1]) automates such chores. It generates an appropriate `typedef` based on a least fixed-point construction, and proves freeness theorems and induction rules, as well as theorems for recursion and case combinators. The user just has to give a simple specification of new inductive types using a notation similar to ML or Haskell.

The current datatype package can handle both mutual and indirect recursion. It also offers to represent existing types as datatypes giving the advantage of a more uniform view on standard theories.

2.6.1 Basics

A general **datatype** definition is of the following form:

$$\begin{aligned} \text{datatype } (\vec{\alpha})t_1 &= C_1^1 \tau_{1,1}^1 \dots \tau_{1,m_1}^1 \mid \dots \mid C_{k_1}^1 \tau_{k_1,1}^1 \dots \tau_{k_1,m_{k_1}}^1 \\ &\vdots \\ \text{and } (\vec{\alpha})t_n &= C_1^n \tau_{1,1}^n \dots \tau_{1,m_1^n}^n \mid \dots \mid C_{k_n}^n \tau_{k_n,1}^n \dots \tau_{k_n,m_{k_n}^n}^n \end{aligned}$$

where $\vec{\alpha} = (\alpha_1, \dots, \alpha_h)$ is a list of type variables, C_i^j are distinct constructor names and $\tau_{i,i'}^j$ are *admissible* types containing at most the type variables $\alpha_1, \dots, \alpha_h$. A type τ occurring in a **datatype** definition is *admissible* if and only if

- τ is non-recursive, i.e. τ does not contain any of the newly defined type constructors t_1, \dots, t_n , or
- $\tau = (\vec{\alpha})t_{j'}$ where $1 \leq j' \leq n$, or
- $\tau = (\tau'_1, \dots, \tau'_{h'})t'$, where t' is the type constructor of an already existing datatype and $\tau'_1, \dots, \tau'_{h'}$ are admissible types.
- $\tau = \sigma \rightarrow \tau'$, where τ' is an admissible type and σ is non-recursive (i.e. the occurrences of the newly defined types are *strictly positive*)

If some $(\vec{\alpha})t_{j'}$ occurs in a type $\tau_{i,i'}^j$ of the form

$$(\dots, \dots (\vec{\alpha})t_{j'} \dots, \dots)t'$$

this is called a *nested* (or *indirect*) occurrence. A very simple example of a datatype is the type **list**, which can be defined by

```
datatype 'a list = Nil
                | Cons 'a ('a list)
```

Arithmetic expressions **aexp** and boolean expressions **bexp** can be modelled by the mutually recursive datatype definition

```
datatype 'a aexp = If_then_else ('a bexp) ('a aexp) ('a aexp)
                | Sum ('a aexp) ('a aexp)
                | Diff ('a aexp) ('a aexp)
                | Var 'a
                | Num nat
and 'a bexp = Less ('a aexp) ('a aexp)
            | And ('a bexp) ('a bexp)
            | Or ('a bexp) ('a bexp)
```

The datatype **term**, which is defined by

```
datatype ('a, 'b) term = Var 'a
                      | App 'b ((( 'a, 'b) term) list)
```

is an example for a datatype with nested recursion. Using nested recursion involving function spaces, we may also define infinitely branching datatypes, e.g.

```
datatype 'a tree = Atom 'a | Branch "nat => 'a tree"
```

Types in HOL must be non-empty. Each of the new datatypes $(\vec{\alpha})t_j$ with $1 \leq j \leq n$ is non-empty if and only if it has a constructor C_i^j with the following property: for all argument types $\tau_{i,i'}^j$ of the form $(\vec{\alpha})t_{j'}$ the datatype $(\vec{\alpha})t_{j'}$ is non-empty.

If there are no nested occurrences of the newly defined datatypes, obviously at least one of the newly defined datatypes $(\vec{\alpha})t_j$ must have a constructor C_i^j without recursive arguments, a *base case*, to ensure that the new types are non-empty. If there are nested occurrences, a datatype can even be non-empty without having a base case itself. Since `list` is a non-empty datatype, `datatype t = C (t list)` is non-empty as well.

Freeness of the constructors

The datatype constructors are automatically defined as functions of their respective type:

$$C_i^j :: [\tau_{i,1}^j, \dots, \tau_{i,m_i^j}^j] \Rightarrow (\alpha_1, \dots, \alpha_h)t_j$$

These functions have certain *freeness* properties. They construct distinct values:

$$C_i^j x_1 \dots x_{m_i^j} \neq C_{i'}^j y_1 \dots y_{m_{i'}^j} \quad \text{for all } i \neq i'.$$

The constructor functions are injective:

$$(C_i^j x_1 \dots x_{m_i^j} = C_i^j y_1 \dots y_{m_i^j}) = (x_1 = y_1 \wedge \dots \wedge x_{m_i^j} = y_{m_i^j})$$

Since the number of distinctness inequalities is quadratic in the number of constructors, the datatype package avoids proving them separately if there are too many constructors. Instead, specific inequalities are proved by a suitable simplification procedure on demand.³

³This procedure, which is already part of the default simpset, may be referred to by the ML identifier `DatatypePackage.distinct_simproc`.

Structural induction

The datatype package also provides structural induction rules. For datatypes without nested recursion, this is of the following form:

$$\begin{array}{c}
 \bigwedge x_1 \dots x_{m_1} . \llbracket P_{s_{1,1}^1} x_{r_{1,1}^1}; \dots; P_{s_{1,l_1^1}^1} x_{r_{1,l_1^1}^1} \rrbracket \implies P_1 (C_1^1 x_1 \dots x_{m_1^1}) \\
 \vdots \\
 \bigwedge x_1 \dots x_{m_{k_1}^1} . \llbracket P_{s_{k_1,1}^1} x_{r_{k_1,1}^1}; \dots; P_{s_{k_1,l_{k_1}^1}^1} x_{r_{k_1,l_{k_1}^1}^1} \rrbracket \implies P_1 (C_{k_1}^1 x_1 \dots x_{m_{k_1}^1}) \\
 \vdots \\
 \bigwedge x_1 \dots x_{m_1^n} . \llbracket P_{s_{1,1}^n} x_{r_{1,1}^n}; \dots; P_{s_{1,l_1^n}^n} x_{r_{1,l_1^n}^n} \rrbracket \implies P_n (C_1^n x_1 \dots x_{m_1^n}) \\
 \vdots \\
 \bigwedge x_1 \dots x_{m_{k_n}^n} . \llbracket P_{s_{k_n,1}^n} x_{r_{k_n,1}^n}; \dots; P_{s_{k_n,l_{k_n}^n}^n} x_{r_{k_n,l_{k_n}^n}^n} \rrbracket \implies P_n (C_{k_n}^n x_1 \dots x_{m_{k_n}^n}) \\
 \hline
 P_1 x_1 \wedge \dots \wedge P_n x_n
 \end{array}$$

where

$$\begin{aligned}
 Rec_i^j &:= \left\{ \left(r_{i,1}^j, s_{i,1}^j \right), \dots, \left(r_{i,l_i^j}^j, s_{i,l_i^j}^j \right) \right\} = \\
 &\quad \left\{ (i', i'') \mid 1 \leq i' \leq m_i^j \wedge 1 \leq i'' \leq n \wedge \tau_{i,i'}^j = (\alpha_1, \dots, \alpha_h) t_{i''} \right\}
 \end{aligned}$$

i.e. the properties P_j can be assumed for all recursive arguments.

For datatypes with nested recursion, such as the `term` example from above, things are a bit more complicated. Conceptually, Isabelle/HOL unfolds a definition like

```

datatype ('a,'b) term = Var 'a
                    | App 'b (((a, 'b) term) list)

```

to an equivalent definition without nesting:

```

datatype ('a,'b) term      = Var
                    | App 'b (('a, 'b) term_list)
and      ('a,'b) term_list = Nil'
                    | Cons' (('a,'b) term) (('a,'b) term_list)

```

Note however, that the type `('a,'b) term_list` and the constructors `Nil'` and `Cons'` are not really introduced. One can directly work with the original (isomorphic) type `((a, 'b) term) list` and its existing constructors `Nil` and `Cons`. Thus, the structural induction rule for `term` gets the form

$$\begin{array}{c}
 \bigwedge x . P_1 (\text{Var } x) \\
 \bigwedge x_1 x_2 . P_2 x_2 \implies P_1 (\text{App } x_1 x_2) \\
 P_2 \text{ Nil} \\
 \bigwedge x_1 x_2 . \llbracket P_1 x_1; P_2 x_2 \rrbracket \implies P_2 (\text{Cons } x_1 x_2) \\
 \hline
 P_1 x_1 \wedge P_2 x_2
 \end{array}$$

Note that there are two predicates P_1 and P_2 , one for the type `('a, 'b) term` and one for the type `(('a, 'b) term) list`.

For a datatype with function types such as `'a tree`, the induction rule is of the form

$$\frac{\bigwedge a. P (\text{Atom } a) \quad \bigwedge ts. (\forall x. P (ts \ x)) \implies P (\text{Branch } ts)}{P \ t}$$

In principle, inductive types are already fully determined by freeness and structural induction. For convenience in applications, the following derived constructions are automatically provided for any datatype.

The case construct

The type comes with an ML-like **case**-construct:

$$\begin{array}{lcl} \text{case } e \text{ of} & C_1^j \ x_{1,1} \ \dots \ x_{1,m_1^j} & \Rightarrow \ e_1 \\ & \vdots & \\ & C_{k_j}^j \ x_{k_j,1} \ \dots \ x_{k_j,m_{k_j}^j} & \Rightarrow \ e_{k_j} \end{array}$$

where the $x_{i,j}$ are either identifiers or nested tuple patterns as in §2.5.1.

! All constructors must be present, their order is fixed, and nested patterns are not supported (with the exception of tuples). Violating this restriction results in strange error messages.

To perform case distinction on a goal containing a **case**-construct, the theorem $t_j.\text{split}$ is provided:

$$\begin{aligned} P(t_j\text{-case } f_1 \ \dots \ f_{k_j} \ e) = & ((\forall x_1 \dots x_{m_1^j}. e = C_1^j \ x_1 \dots x_{m_1^j} \rightarrow P(f_1 \ x_1 \dots x_{m_1^j})) \\ & \wedge \dots \wedge \\ & (\forall x_1 \dots x_{m_{k_j}^j}. e = C_{k_j}^j \ x_1 \dots x_{m_{k_j}^j} \rightarrow P(f_{k_j} \ x_1 \dots x_{m_{k_j}^j}))) \end{aligned}$$

where $t_j\text{-case}$ is the internal name of the **case**-construct. This theorem can be added to a simpset via **addsplits** (see §2.4.1).

Case splitting on assumption works as well, by using the rule $t_j.\text{split_asm}$ in the same manner. Both rules are available under $t_j.\text{splits}$ (this name is *not* bound in ML, though).

! By default only the selector expression (e above) in a **case**-construct is simplified, in analogy with **if** (see page 20). Only if that reduces to a constructor is one of the arms of the **case**-construct exposed and simplified. To ensure full simplification of all parts of a **case**-construct for datatype t , remove $t.\text{case_weak_cong}$ from the simpset, for example by **delcongs** `[thm "t.weak_case_cong"]`.

The function `size`

Theory `NatArith` declares a generic function `size` of type $\alpha \Rightarrow \text{nat}$. Each datatype defines a particular instance of `size` by overloading according to the following scheme:

$$\text{size}(C_i^j \ x_1 \ \dots \ x_{m_i^j}) = \begin{cases} 0 & \text{if } \text{Rec}_i^j = \emptyset \\ 1 + \sum_{h=1}^{l_i^j} \text{size } x_{r_{i,h}^j} & \text{if } \text{Rec}_i^j = \left\{ (r_{i,1}^j, s_{i,1}^j), \dots, (r_{i,l_i^j}^j, s_{i,l_i^j}^j) \right\} \end{cases}$$

where Rec_i^j is defined above. Viewing datatypes as generalised trees, the size of a leaf is 0 and the size of a node is the sum of the sizes of its subtrees + 1.

2.6.2 Defining datatypes

The theory syntax for datatype definitions is given in the Isabelle/Isar reference manual. In order to be well-formed, a datatype definition has to obey the rules stated in the previous section. As a result the theory is extended with the new types, the constructors, and the theorems listed in the previous section.

Most of the theorems about datatypes become part of the default simpset and you never need to see them again because the simplifier applies them automatically. Only induction or case distinction are usually invoked by hand.

`induct_tac "x" i` applies structural induction on variable x to subgoal i , provided the type of x is a datatype.

`induct_tac "x1 ... xn" i` applies simultaneous structural induction on the variables x_1, \dots, x_n to subgoal i . This is the canonical way to prove properties of mutually recursive datatypes such as `aexp` and `bexp`, or datatypes with nested recursion such as `term`.

In some cases, induction is overkill and a case distinction over all constructors of the datatype suffices.

`case_tac "u" i` performs a case analysis for the term u whose type must be a datatype. If the datatype has k_j constructors $C_1^j, \dots, C_{k_j}^j$, subgoal i is replaced by k_j new subgoals which contain the additional assumption $u = C_{i'}^j \ x_1 \ \dots \ x_{m_{i'}^j}$ for $i' = 1, \dots, k_j$.

Note that induction is only allowed on free variables that should not occur among the premises of the subgoal. Case distinction applies to arbitrary terms.

For the technically minded, we exhibit some more details. Processing the theory file produces an ML structure which, in addition to the usual components, contains a structure named *t* for each datatype *t* defined in the file. Each structure *t* contains the following elements:

```
val distinct : thm list
val inject   : thm list
val induct   : thm
val exhaust  : thm
val cases    : thm list
val split    : thm
val split_asm : thm
val recs     : thm list
val size     : thm list
val simps    : thm list
```

`distinct`, `inject`, `induct`, `size` and `split` contain the theorems described above. For user convenience, `distinct` contains inequalities in both directions. The reduction rules of the `case`-construct are in `cases`. All theorems from `distinct`, `inject` and `cases` are combined in `simps`. In case of mutually recursive datatypes, `recs`, `size`, `induct` and `simps` are contained in a separate structure named $t_1 \dots t_n$.

2.7 Old-style recursive function definitions

Old-style recursive definitions via `recdef` requires that you supply a well-founded relation that governs the recursion. Recursive calls are only allowed if they make the argument decrease under the relation. Complicated recursion forms, such as nested recursion, can be dealt with. Termination can even be proved at a later time, though having unsolved termination conditions around can make work difficult.⁴

Using `recdef`, you can declare functions involving nested recursion and pattern-matching. Recursion need not involve datatypes and there are few syntactic restrictions. Termination is proved by showing that each recursive call makes the argument smaller in a suitable sense, which you specify by supplying a well-founded relation.

Here is a simple example, the Fibonacci function. The first line declares `fib` to be a constant. The well-founded relation is simply `<` (on the natural numbers). Pattern-matching is used here: `1` is a macro for `Suc 0`.

⁴This facility is based on Konrad Slind's TFL package [4]. Thanks are due to Konrad for implementing TFL and assisting with its installation.


```

consts fib  :: "nat => nat"
recdef fib "less_than"
  "fib 0 = 0"
  "fib 1 = 1"
  "fib (Suc(Suc x)) = (fib x + fib (Suc x))"

```

With `recdef`, function definitions may be incomplete, and patterns may overlap, as in functional programming. The `recdef` package disambiguates overlapping patterns by taking the order of rules into account. For missing patterns, the function is defined to return a default value.

The well-founded relation defines a notion of “smaller” for the function’s argument type. The relation \prec is **well-founded** provided it admits no infinitely decreasing chains

$$\cdots \prec x_n \prec \cdots \prec x_1.$$

If the function’s argument has type τ , then \prec has to be a relation over τ : it must have type $(\tau \times \tau) \text{set}$.

Proving well-foundedness can be tricky, so Isabelle/HOL provides a collection of operators for building well-founded relations. The package recognises these operators and automatically proves that the constructed relation is well-founded. Here are those operators, in order of importance:

- `less_than` is “less than” on the natural numbers. (It has type $(\text{nat} \times \text{nat}) \text{set}$, while $<$ has type $[\text{nat}, \text{nat}] \Rightarrow \text{bool}$.)
- `measure f`, where f has type $\tau \Rightarrow \text{nat}$, is the relation \prec on type τ such that $x \prec y$ if and only if $f(x) < f(y)$. Typically, f takes the recursive function’s arguments (as a tuple) and returns a result expressed in terms of the function `size`. It is called a **measure function**. Recall that `size` is overloaded and is defined on all datatypes (see §2.6.1).
- `inv_image R f` is a generalisation of `measure`. It specifies a relation such that $x \prec y$ if and only if $f(x)$ is less than $f(y)$ according to R , which must itself be a well-founded relation.
- `R1<*lex*>R2` is the lexicographic product of two relations. It is a relation on pairs and satisfies $(x_1, x_2) \prec (y_1, y_2)$ if and only if x_1 is less than y_1 according to R_1 or $x_1 = y_1$ and x_2 is less than y_2 according to R_2 .
- `finite_psubset` is the proper subset relation on finite sets.

We can use `measure` to declare Euclid’s algorithm for the greatest common divisor. The measure function, $\lambda(m, n) . n$, specifies that the recursion terminates because argument n decreases.

```

recdef gcd "measure ((%(m,n). n) :: nat*nat=>nat)"
  "gcd (m, n) = (if n=0 then m else gcd(n, m mod n))"

```

The general form of a well-founded recursive definition is

```

recdef function rel
  congs      congruence rules      (optional)
  simpset    simplification set    (optional)
  reduction rules

```

where

- *function* is the name of the function, either as an *id* or a *string*.
- *rel* is a HOL expression for the well-founded termination relation.
- *congruence rules* are required only in highly exceptional circumstances.
- The *simplification set* is used to prove that the supplied relation is well-founded. It is also used to prove the **termination conditions**: assertions that arguments of recursive calls decrease under *rel*. By default, simplification uses `simpset()`, which is sufficient to prove well-foundedness for the built-in relations listed above.
- *reduction rules* specify one or more recursion equations. Each left-hand side must have the form $f\ t$, where f is the function and t is a tuple of distinct variables. If more than one equation is present then f is defined by pattern-matching on components of its argument whose type is a **datatype**.

The ML identifier `f.simps` contains the reduction rules as a list of theorems.

With the definition of `gcd` shown above, Isabelle/HOL is unable to prove one termination condition. It remains as a precondition of the recursion theorems:

```

gcd.simps;
[!" m n. n ~= 0 --> m mod n < n
  ==> gcd (?m, ?n) = (if ?n=0 then ?m else gcd (?n, ?m mod ?n))"]
: thm list

```

The theory `HOL/ex/Primes` illustrates how to prove termination conditions afterwards. The function `Tfl.tgoalw` is like the standard function `goalw`, which sets up a goal to prove, but its argument should be the identifier `f.simps` and its effect is to set up a proof of the termination conditions:

```
Tf1.tgoalw thy [] gcd.simps;
Level 0
! m n. n ~= 0 --> m mod n < n
1. ! m n. n ~= 0 --> m mod n < n
```

This subgoal has a one-step proof using `simp_tac`. Once the theorem is proved, it can be used to eliminate the termination conditions from elements of `gcd.simps`. Theory `HOL/Subst/Unify` is a much more complicated example of this process, where the termination conditions can only be proved by complicated reasoning involving the recursive function itself.

Isabelle/HOL can prove the `gcd` function's termination condition automatically if supplied with the right `simpset`.

```
recdef gcd "measure ((%(m,n). n) :: nat*nat=>nat)"
simpset "simpset() addsimps [mod_less_divisor, zero_less_eq]"
"gcd (m, n) = (if n=0 then m else gcd(n, m mod n))"
```

If all termination conditions were proved automatically, `f.simps` is added to the `simpset` automatically, just as in `primrec`. The simplification rules corresponding to clause `i` (where counting starts at 0) are called `f.i` and can be accessed as `thms "f.i"`, which returns a list of theorems. Thus you can, for example, remove specific clauses from the `simpset`. Note that a single clause may give rise to a set of simplification rules in order to capture the fact that if clauses overlap, their order disambiguates them.

A `recdef` definition also returns an induction rule specialised for the recursive function. For the `gcd` function above, the induction rule is

```
gcd.induct;
"(!m n. n ~= 0 --> ?P n (m mod n) ==> ?P m n) ==> ?P ?u ?v" : thm
```

This rule should be used to reason inductively about the `gcd` function. It usually makes the induction hypothesis available at all recursive calls, leading to very direct proofs. If any termination conditions remain unproved, they will become additional premises of this rule.

2.8 Example: Cantor's Theorem

Cantor's Theorem states that every set has more subsets than it has elements. It has become a favourite example in higher-order logic since it is so easily expressed:

$$\forall f :: \alpha \Rightarrow \alpha \Rightarrow \text{bool} . \exists S :: \alpha \Rightarrow \text{bool} . \forall x :: \alpha . f\ x \neq S$$

Viewing types as sets, $\alpha \Rightarrow \text{bool}$ represents the powerset of α . This version states that for every function from α to its powerset, some subset is outside its range.

The Isabelle proof uses HOL's set theory, with the type $\alpha \text{ set}$ and the operator `range`.

```
context Set.thy;
```

The set S is given as an unknown instead of a quantified variable so that we may inspect the subset found by the proof.

```
Goal "?S ~: range (f :: 'a=>'a set)";
  Level 0
  ?S ~: range f
  1. ?S ~: range f
```

The first two steps are routine. The rule `rangeE` replaces $?S \in \text{range } f$ by $?S = f \ x$ for some x .

```
by (resolve_tac [notI] 1);
  Level 1
  ?S ~: range f
  1. ?S : range f ==> False
by (eresolve_tac [rangeE] 1);
  Level 2
  ?S ~: range f
  1. !!x. ?S = f x ==> False
```

Next, we apply `equalityCE`, reasoning that since $?S = f \ x$, we have $?c \in ?S$ if and only if $?c \in f \ x$ for any $?c$.

```
by (eresolve_tac [equalityCE] 1);
  Level 3
  ?S ~: range f
  1. !!x. [| ?c3 x : ?S; ?c3 x : f x |] ==> False
  2. !!x. [| ?c3 x ~: ?S; ?c3 x ~: f x |] ==> False
```

Now we use a bit of creativity. Suppose that $?S$ has the form of a comprehension. Then $?c \in \{x. ?P \ x\}$ implies $?P \ ?c$. Destruct-resolution using `CollectD` instantiates $?S$ and creates the new assumption.

```
by (dresolve_tac [CollectD] 1);
  Level 4
  {x. ?P7 x} ~: range f
  1. !!x. [| ?c3 x : f x; ?P7(?c3 x) |] ==> False
  2. !!x. [| ?c3 x ~: {x. ?P7 x}; ?c3 x ~: f x |] ==> False
```

Forcing a contradiction between the two assumptions of subgoal 1 completes the instantiation of S . It is now the set $\{x. x \notin f \ x\}$, which is the standard diagonal construction.

```

by (contr_tac 1);
Level 5
{x. x ~: f x} ~: range f
1. !!x. [| x ~: {x. x ~: f x}; x ~: f x |] ==> False

```

The rest should be easy. To apply `CollectI` to the negated assumption, we employ `swap_res_tac`:

```

by (swap_res_tac [CollectI] 1);
Level 6
{x. x ~: f x} ~: range f
1. !!x. [| x ~: f x; ~ False |] ==> x ~: f x
by (assume_tac 1);
Level 7
{x. x ~: f x} ~: range f
No subgoals!

```

How much creativity is required? As it happens, Isabelle can prove this theorem automatically. The default classical set `claset()` contains rules for most of the constructs of HOL's set theory. We must augment it with `equalityCE` to break up set equalities, and then apply best-first search. Depth-first search would diverge, but best-first search successfully navigates through the large search space.

```

choplev 0;
Level 0
?S ~: range f
1. ?S ~: range f
by (best_tac (claset() addSEs [equalityCE]) 1);
Level 1
{x. x ~: f x} ~: range f
No subgoals!

```

If you run this example interactively, make sure your current theory contains theory `Set`, for example by executing `context Set.thy`. Otherwise the default `claset` may not contain the rules for set theory.

Bibliography

- [1] Stefan Berghofer and Markus Wenzel. Inductive datatypes in HOL — lessons learned in Formal-Logic Engineering. In Y. Bertot, G. Dowek, A. Hirschowitz, C. Paulin, and L. Thery, editors, *Theorem Proving in Higher Order Logics: TPHOLs '99*, volume 1690 of *Lecture Notes in Computer Science*. Springer-Verlag, 1999.
- [2] M. J. C. Gordon and T. F. Melham, editors. *Introduction to HOL: A Theorem Proving Environment for Higher Order Logic*. Cambridge University Press, 1993.
- [3] Lawrence C. Paulson. A formulation of the simple theory of types (for Isabelle). In P. Martin-Löf and G. Mints, editors, *COLOG-88: International Conference on Computer Logic*, LNCS 417, pages 246–274, Tallinn, Published 1990. Estonian Academy of Sciences, Springer.
- [4] Konrad Slind. Function definition in higher order logic. In J. von Wright, J. Grundy, and J. Harrison, editors, *Theorem Proving in Higher Order Logics: TPHOLs '96*, volume 1125 of *Lecture Notes in Computer Science*, pages 381–397. Springer-Verlag, 1996.

Index

- ! symbol, 4, 6, 13, 15, 28
- | symbol, 4
- [] symbol, 28
- # symbol, 28
- & symbol, 4
- * symbol, 3, 25
- * type, 23
- + symbol, 3, 25
- + type, 23
- symbol, 3, 25
- > symbol, 4
- : symbol, 12
- < constant, 24
- < symbol, 25
- <= constant, 24
- <= symbol, 12
- = symbol, 4
- ? symbol, 4, 6, 15
- ?! symbol, 4
- @ symbol, 4, 28
- ^ symbol, 3
- ‘ ‘ symbol, 12
- { } symbol, 12

- 0 constant, 3, 25

- Addsplits, 21**
- addsplits, 21, 30, 34**
- ALL** symbol, 4, 13, 15
- All** constant, 4
- All_def** theorem, 9
- all_dupE** theorem, 11
- allE** theorem, 11
- allI** theorem, 11
- and_def** theorem, 9
- arg_cong** theorem, 10

- Ball** constant, 12, 15
- Ball_def** theorem, 15
- ballE** theorem, 16
- ballI** theorem, 16
- Bex** constant, 12, 15
- Bex_def** theorem, 15
- bexCI** theorem, 16, 18
- bexE** theorem, 16
- bexI** theorem, 16, 18
- bool* type, 3
- box_equals** theorem, 9, 10
- bspec** theorem, 16
- butlast** constant, 28

- case** symbol, 26, 27, 34
- case_tac, 9, 35**
- case_weak_cong, 34**
- ccontr** theorem, 11
- classical** theorem, 11
- Collect** constant, 12, 14
- Collect_mem_eq** theorem, 15
- CollectD** theorem, 16, 40
- CollectE** theorem, 16
- CollectI** theorem, 16, 41
- Compl_def** theorem, 15
- Compl_disjoint** theorem, 19
- Compl_Int** theorem, 19
- Compl_partition** theorem, 19
- Compl_Un** theorem, 19
- ComplD** theorem, 17
- ComplI** theorem, 17
- concat** constant, 28
- cong** theorem, 10

- conj_cong, 20
- conjE theorem, 10
- conjI theorem, 10
- conjunct1 theorem, 10
- conjunct2 theorem, 10
- context, 41
- datatype, 30
- Delsplits, **21**
- delsplits, **21**
- disjCI theorem, 11
- disjE theorem, 10
- disjI1 theorem, 10
- disjI2 theorem, 10
- div symbol, 25
- div_geq theorem, 25
- div_less theorem, 25
- Divides theory, 26
- double_complement theorem, 19
- drop constant, 28
- dropWhile constant, 28
- dvd symbol, 25
- empty_def theorem, 15
- emptyE theorem, 17
- Eps constant, 4, 6
- equalityCE theorem, 16, 18, 40, 41
- equalityD1 theorem, 16
- equalityD2 theorem, 16
- equalityE theorem, 16
- equalityI theorem, 16
- EX symbol, 4, 13, 15
- Ex constant, 4
- EX! symbol, 4
- Ex1 constant, 4
- Ex1_def theorem, 9
- ex1E theorem, 11
- ex1I theorem, 11
- Ex_def theorem, 9
- exCI theorem, 11
- excluded_middle theorem, 11
- exE theorem, 11
- exI theorem, 11
- ext theorem, 8
- False constant, 4
- False_def theorem, 9
- FalseE theorem, 10
- filter constant, 28
- foldl constant, 28
- fst constant, 22
- fst_conv theorem, 22
- Fun theory, 19
- fun type, 3
- fun_cong theorem, 10
- hd constant, 28
- higher-order logic, 3–41
- HOL, **7**
- HOL system, 3, 6
- HOL_basic_ss, **20**
- HOL_ss, **20**
- hyp_subst_tac, 20
- If constant, 4
- if, 20
- if_def theorem, 9
- if_not_P theorem, 11
- if_P theorem, 11
- if_weak_cong, 20
- iff theorem, 8
- iffCE theorem, 11, 18
- iffD1 theorem, 10
- iffD2 theorem, 10
- iffE theorem, 10
- iffI theorem, 10
- image_def theorem, 15
- imageE theorem, 17
- imageI theorem, 17
- impCE theorem, 11
- impE theorem, 10
- impI theorem, 8
- in symbol, 5

- ind* type, 24
- induct_tac*, 26, **35**
- inj* constant, 19
- inj_def* theorem, 19
- inj_Inl* theorem, 24
- inj_Inr* theorem, 24
- inj_on* constant, 19
- inj_on_def* theorem, 19
- inj_Suc* theorem, 25
- Inl* constant, 24
- Inl_not_Inr* theorem, 24
- Inr* constant, 24
- insert* constant, 12
- insert_def* theorem, 15
- insertE* theorem, 17
- insertI1* theorem, 17
- insertI2* theorem, 17
- INT* symbol, 12, 13, 15
- Int* symbol, 12
- int* theorem, 3, 6, 26
- Int_absorb* theorem, 19
- Int_assoc* theorem, 19
- Int_commute* theorem, 19
- INT_D* theorem, 17
- Int_def* theorem, 15
- INT_E* theorem, 17
- Int_greatest* theorem, 18
- INT_I* theorem, 17
- Int_lower1* theorem, 18
- Int_lower2* theorem, 18
- Int_Un_distrib* theorem, 19
- Int_Union* theorem, 19
- IntD1* theorem, 17
- IntD2* theorem, 17
- IntE* theorem, 17
- INTER* constant, 12
- Inter* constant, 12
- INTER1* constant, 12
- INTER1_def* theorem, 15
- INTER_def* theorem, 15
- Inter_def* theorem, 15
- Inter_greatest* theorem, 18
- Inter_lower* theorem, 18
- Inter_Un_distrib* theorem, 19
- InterD* theorem, 17
- InterE* theorem, 17
- InterI* theorem, 17
- IntI* theorem, 17
- inv* constant, 19
- inv_def* theorem, 19
- last* constant, 28
- LEAST* constant, 6, 7, 24
- Least* constant, 4
- Least_def* theorem, 9
- length* constant, 28
- less_induct* theorem, 26
- Let* constant, 4, 7
- let* symbol, 5
- Let_def* theorem, 7, 9
- Lin_Arith.tac*, 27
- linorder* class, 6, 26
- List* theory, 27, 28
- list* type, 27–30
- map* constant, 28
- max* constant, 6, 24
- mem* symbol, 28
- mem_Collect_eq* theorem, 15
- min* constant, 6, 24
- minus* class, 3
- mod* symbol, 25
- mod_geq* theorem, 25
- mod_less* theorem, 25
- mono* constant, 6
- mp* theorem, 8
- n_not_Suc_n* theorem, 25
- Nat* theory, 24, 26
- nat* type, 24–26
- nat* type, 24–27
- nat* theorem, 3, 6
- nat_induct* theorem, 25

- nat_rec constant, 26
- NatArith theory, 26
- Not constant, 4
- not_def theorem, 9
- not_sym theorem, 10
- notE theorem, 10
- notI theorem, 10
- notnotD theorem, 11
- null constant, 28
- o symbol, 4, 18
- o_def theorem, 9
- of symbol, 7
- or_def theorem, 9
- Ord theory, 6
- ord class, 6, 7, 24
- order class, 6
- Pair constant, 22
- Pair_eq theorem, 22
- Pair_inject theorem, 22
- PairE theorem, 22
- plus class, 3
- plus_ac0 class, 6
- Pow constant, 12
- Pow_def theorem, 15
- PowD theorem, 17
- power class, 3
- PowI theorem, 17
- primrec symbol, 26
- priorities, 1
- Prod theory, 23
- range constant, 12, 40
- range_def theorem, 15
- rangeE theorem, 17, 40
- rangeI theorem, 17
- real theorem, 3, 6, 26
- recdef, 36–39
- recursion
 - general, 36–39
- refl theorem, 8
- res_inst_tac, 6
- rev constant, 28
- search
 - best-first, 41
- Set theory, 14, 15
- set constant, 28
- set type, 14
- set_diff_def theorem, 15
- setsum constant, 6
- show_sorts, 6
- show_types, 6
- Sigma constant, 22
- Sigma_def theorem, 22
- SigmaE theorem, 22
- SigmaI theorem, 22
- simplification
 - of case, 34
 - of if, 20
 - of conjunctions, 20
- size constant, 35
- smp_tac, 9
- snd constant, 22
- snd_conv theorem, 22
- SOME symbol, 4
- some_equality theorem, 8, 11
- someI theorem, 8
- spec theorem, 11
- split constant, 22
- split theorem, 22
- split_all_tac, **23**
- split_if theorem, 11, 21
- split_list_case theorem, 27
- split_split theorem, 22
- ssubst theorem, 9, 10
- stac, **20**
- strip_tac, 9
- subset_def theorem, 15
- subset_refl theorem, 16
- subset_trans theorem, 16
- subsetCE theorem, 16, 18

- subsetD theorem, 16, 18
- subsetI theorem, 16
- subst theorem, 8
- Suc constant, 25
- Suc_not_Zero theorem, 25
- Sum theory, 24
- sum.split_case theorem, 24
- sum_case constant, 24
- sum_case_Inl theorem, 24
- sum_case_Inr theorem, 24
- sumE theorem, 24
- surj constant, 18, 19
- surj_def theorem, 19
- surjective_pairing theorem, 22
- surjective_sum theorem, 24
- swap theorem, 11
- swap_res_tac, 41
- sym theorem, 10
- take constant, 28
- takeWhile constant, 28
- term class, 3
- times class, 3
- tl constant, 28
- tracing
 - of unification, 6
- trans theorem, 10
- True constant, 4
- True_def theorem, 9
- True_or_False theorem, 8
- TrueI theorem, 10
- Trueprop constant, 4
- UN symbol, 12, 13, 15
- Un symbol, 12
- Un1 theorem, 18
- Un2 theorem, 18
- Un_absorb theorem, 19
- Un_assoc theorem, 19
- Un_commute theorem, 19
- Un_def theorem, 15
- UN_E theorem, 17
- UN_I theorem, 17
- Un_Int_distrib theorem, 19
- Un_Inter theorem, 19
- Un_least theorem, 18
- Un_upper1 theorem, 18
- Un_upper2 theorem, 18
- UnCI theorem, 17, 18
- UnE theorem, 17
- UnI1 theorem, 17
- UnI2 theorem, 17
- unification
 - incompleteness of, 6
- Unify.trace_types, 6
- UNION constant, 12
- Union constant, 12
- UNION1 constant, 12
- UNION1_def theorem, 15
- UNION_def theorem, 15
- Union_def theorem, 15
- Union_least theorem, 18
- Union_Un_distrib theorem, 19
- Union_upper theorem, 18
- UnionE theorem, 17
- UnionI theorem, 17
- unit_eq theorem, 24